Chapter 10. Multiple Turtles

"Wow…working with one turtle is bad enough. But MSW Logo offers 1024 turtles!

"That's worse than working with rabbits!"



Yes, there are lots of turtles in MSW Logo…1,024 to be exact. But before getting busy with multiple turtles, take a look at how you can simulate multiple turtles. It's a great review of things you've been doing up until now.

The KALEIDOSCOPE procedure shows a good use of coordinate commands. The resulting picture looks as if it was drawn by four turtles.

```
TO KALEIDOSCOPE
SETUP
REPEAT 50 [DEMO]
END
```

```
TO DEMO
IF :STEP > 100 [MAKE “STEP 10]
MAKE “STEP :STEP + 5 MOVE RT :ANGLE
IF OR :ANGLE > 45 :ANGLE < -45 [MAKE :ANGLE
      :ANGLE - ( :ANGLE * 2 ) ]
MAKE “ANGLE :ANGLE + 5
IF :C = 3 [MAKE “C 0]
MAKE “C :C + 1
END

TO MOVE
MAKE “X1 XCOR MAKE “Y1 YCOR FD :STEP
MAKE “X2 XCOR MAKE “Y2 YCOR
PU SETXY :X1 - ( :X1 * 2 ) :Y1
PD SETXY :X2 - ( :X2 * 2 ) :Y2
PU SETXY :X1 - ( :X1 * 2 ) :Y1 - ( :y1 * 2 )
PD SETXY :X2 - ( :X2 * 2 ) :Y2 - ( :Y2 * 2 )
PU SETXY :X1 :Y1 - ( :Y1 * 2 )
PD SETXY :X2 :Y2 - ( :Y2 * 2 )
PU SETXY :X2 :Y2 PD RT 15
END

TO SETUP
MAKE “C 1
MAKE “STEP 20
MAKE “ANGLE 5 HT
END
```



Imagine what this would look like if it was drawn using different colors.

Here's another good example of using coordinate commands in a procedure. In the one above, the "turtles" all acted according to plan. In the one shown on the next page, you tell each one what to do.



Four "turtles" are defined in this procedure even though only one is used. The SETUP procedure defines each "turtle" by defining its position and heading. XCOR, YCOR, and HEADING are each spelled out for each turtle so you can see how this procedure works.

Type START to begin. Then ASK a turtle…0, 1, 2, or 3…to do something. You'll notice that ASK requires a turtle number (:TNUM) and a :COMMAND.LIST. So be sure to enclose your instructions to the turtle in brackets.

```
ASK 2 [REPEAT 4 [FD 100 RT 90]]

TO ASK :TNUM :COMMAND.LIST
IF :TNUM = 0 [ZERO :COMMAND.LIST]
IF :TNUM = 1 [ONE :COMMAND.LIST]
IF :TNUM = 2 [TWO :COMMAND.LIST]
IF :TNUM = 3 [THREE :COMMAND.LIST]
END

TO ONE  :COMMAND.LIST
SETXY :OLDX1 :OLDY1 SETH :OLDH1
PD RUN :COMMAND.LIST PU
MAKE “OLDX1 XCOR
MAKE “OLDY1 YCOR
MAKE “OLDH1 HEADING
END

TO SET.ONE
MAKE “OLDX1 XCOR
MAKE “OLDY1 YCOR
MAKE “OLDH1 HEADING
END

TO SET.THREE
MAKE “OLDX3 XCOR
MAKE “OLDY3 YCOR
MAKE “OLDH3 HEADING
END

TO SET.TWO
MAKE “OLDX2 XCOR
MAKE “OLDY2 YCOR
MAKE “OLDH2 HEADING
END
```

```
TO SET.ZERO
MAKE “OLDX0 XCOR
MAKE “OLDY0 YCOR
MAKE “OLDH0 HEADING
END

TO SETUP
SET.ZERO
SET.ONE
SET.TWO
SET.THREE
END

TO START
CT PRINT [Welcome to the illusion of multiple turtles!]
TIMER
PRINT [This procedure lets you play with four]
PRINT [turtles: 0, 1, 2, and 3.  Just “ASK”]
PRINT [them TO do what you want each TO do.]
TIMER
PRINT [For example...]
PRINT [ASK 0 [REPEAT 3 [FD 100 RT 120]]] TIMER
PRINT “
PRINT [Ready?  OK, give it a try...]
SETUP
END

TO THREE :COMMAND.LIST
SETXY :OLDX3 :OLDY3 SETH :OLDH3
PD RUN :COMMAND.LIST PU
MAKE “OLDX3 XCOR
MAKE “OLDY3 YCOR
MAKE “OLDH3 HEADING
END
```

```
TO TIMER
WAIT 150 CT
END

TO TWO :COMMAND.LIST
SETXY :OLDX2 :OLDY2 SETH :OLDH2
PD RUN :COMMAND.LIST PU
MAKE “OLDX2 XCOR
MAKE “OLDY2 YCOR
MAKE “OLDH2 HEADING
END

TO ZERO :COMMAND.LIST
SETXY :OLDX0 :OLDY0 SETH :OLDH0
PD RUN :COMMAND.LIST PU
MAKE “OLDX0 XCOR
MAKE “OLDY0 YCOR
MAKE “OLDH0 HEADING
END
```

OK, but how do you go about doing that?

That’s easy enough. In MSW Logo, you simply have to use the command, SETTURTLE.

Since TELL is shorter and easier to understand, let’s write a simple TELL procedure.

```
TO TELL :TNUM
SETTURTLE :TNUM
END
```

We started this book with a race. So how about a simple turtle race?

```
TO RACE
SETUP
TELL RANDOM 4 FD RANDOM 10
TELL 0 IF YCOR = 300 [PRINT [Turtle 0 is the winner!]
     HALT]
TELL 1 IF YCOR = 300 [PRINT [Turtle 1 is the winner!]
     HALT]
TELL 2 IF YCOR = 300 [PRINT [Turtle 2 is the winner!]
     HALT]
TELL 3 IF YCOR = 300 [PRINT [Turtle 3 is the winner!]
     HALT]
RACE
END

TO SETUP
TELL 0 PU SETPOS [-100 -100]
TELL 1 PU SETPOS [-50 -100]
TELL 2 PU SETPOS [0 -100]
TELL 3 PU SETPOS [50 -100]
END
```

This race procedure is very simple. You might want to add a race track including a start and finish line.

A New Target Game

Here's a new target game. You get the chance to hit a moving target.

```
TO SKEET
CT
PRINT [Welcome to the game of...]
PRINT [* * * * * ZAP THE TURTLE * * * * *]
PRINT "
PRINT [One of the Turtles will appear on the]
PRINT [screen. Can you guess the proper direction]
PRINT [and speed to hit the moving Turtle?]
PRINT "
WAIT 200 CT
PRINT [Aw, C'mon!] WAIT 50
PRINT [Give it a try!] WAIT 50
PRINT [To play, press 'Z and then Enter.]
END

TO Z
CS CT PU
MAKE "ANS1 ( RANDOM 500 ) - ( RANDOM 250 )
MAKE "ANS2 ( RANDOM 200 ) - ( RANDOM 100 )
TELL 1 HOME TELL 0 PU SETXY :ANS1 :ANS2 PD
SETH (RANDOM 180) - (RANDOM 45)
IF HEADING < 45 [SETH HEADING + 45]
MAKE "DIR RANDOM 2
IF :DIR = 1 [SETH HEADING - 180]
PRINT [Can you hit the target?]
WAIT 80 CT
SETUP
END
```

```
TO SETUP
REPEAT 3 [TELL 0 FD RANDOM 25]
PRINT [Guess the Heading to the target?]
MAKE “ANS3 READNUMBER
TELL 1 SETH :ANS3
PRINT [Guess the speed to intersect the target?]
MAKE “ANS4 READNUMBER
WRAP
MOVE
END

TO MOVE
TELL 0 FD RANDOM 10
MAKE “ANS1 XCOR
MAKE “ANS2 YCOR
TELL 1 FD :ANS4
MAKE “ANS5 XCOR
MAKE “ANS6 YCOR
IF :ANS5 < :ANS1 - 10 [MOVE]
IF :ANS5 > :ANS1 + 10 [MOVE]
IF :ANS6 < :ANS2 - 10 [MOVE]
IF :ANS6 > :ANS2 + 10 [MOVE]
CHEERS
END

TO READNUMBER
OUTPUT FIRST RL
END

TO TELL :TNUM
SETTURTLE :TNUM
END
```

```
TO CHEERS
CT
REPEAT 5 [PRINT [CONGRATULATIONS!]]
T
END
```

This is just a beginning of what you can do with this game. There's lots of things you can do to dress it up.

First, let's take it apart.

The ZAP procedure get's you started. It tell's you what you have to do.

To start the actual game, press "Z."

The Z procedure sets up the game. Turtle 0 is put in a random position on the screen. It's pen is put down so that it will draw a short line to show you the direction in which it's moving.

Your job will be to guess the direction and speed of your turtle to intercept the first turtle. It's going to keep moving across the screen in the direction it's heading.

You have to set the heading to intercept Turtle 0. And you have to set the speed. You'll get used to the speed by trial and error. The higher the speed, the greater the chance for error. Keep it down in the 5 to 20 range.

Once you enter the direction and speed, the computer takes over. If you come within 10 turtle steps of Turtle 0, you'll get the CHEERS procedure. Otherwise, you'll have to HALT the procedure and try again.

Read the procedures carefully. Then try this game a few times.

What can you do to make it easier?

How would you make it harder?



What would you do to make the CHEERS procedure a bit flashier. Printing Congratulations is a bit dull.

There's lots of things you can do to make this game better. Go do it!

What about changing the shape of the turtle…using a BITMAP to change the shape of the turtle?

Checkout the MSW Logo demo. Notice how the turtle is changed to a bitmap image. Don't get confused by that term 'Bitmap. All it means is that you are using a graphic image, just as the triangle is a graphic image.

Also, read the section of the MSW Logo online help about BITMAP FUNCTIONS. This gives you lots of ideas. Among other things, it explains what was done in the MSW Logo demo.

Here are two examples of bitmaps you can use as turtles. You can make them bigger or smaller depending on how they are to be used.

```
TO BODY
FD 5 RT 90 FD 5 LT 90
REPEAT 2 [FD 30 RT 90 FD 10 RT 90]
RT 90 FD 10 LT 90
END

TO CAR
CS HT WHEEL RWING BODY
WHEEL FD 20 WHEEL
PU SETX XCOR - 15 PD
WHEEL FD 10 FWING
COCKPIT
END

TO COCKPIT
PU SETX XCOR + 8 SETY YCOR - 24 PD
REPEAT 2 [FD 12 RT 90 FD 4 RT 90]
FILLUP
END

TO FILLUP
FD 2 RT 90 FD 2
SETFC [000 000 000] FILL
BK 2 LT 90 BK 2
END

TO FWING
REPEAT 2 [FD 5 RT 90 FD 20 RT 90]
END

TO RWING
BK 5
REPEAT 2 [FD 5 RT 90 FD 20 RT 90]
END
```

```
TO WHEEL
REPEAT 2 [FD 10 RT 90 FD 5 RT 90]
FILLUP
END
```

What about a jet aircraft?



```
TO PLANE
CS HT FD 15 BK 15 LT 90
FD 15 RT 90 FD 5
RT 60 FD 8 SETH 0
FD 15 LT 120 FD 20
SETH 0 FD 5 SETH 30
FD 35 SETH 15 FD 30
PU HOME PD RT 90
FD 15 LT 90 FD 5
LT 60 FD 8 SETH 0
FD 15 RT 120 FD 20
SETH 0 FD 5 SETH 330
FD 35 SETH 345 FD 30
COCKPIT
END

TO COCKPIT
PU SETXY -2 30 PD
REPEAT 2 [FD 12 RT 90 FD 4 RT 90]
FILLUP
END

TO FILLUP
FD 2 RT 90 FD 2
SETFC [000 000 000] FILL
BK 2 LT 90 BK 2
END
```



You’re getting pretty good at this stuff now. So why not try a few other things using your own using multiple turtle procedures?